

Dział „Projekty Czytelników” zawiera opisy projektów nadesłanych do redakcji EP przez Czytelników. Redakcja nie bierze odpowiedzialności za prawidłowe działanie opisywanych układów, gdyż nie testujemy ich laboratoryjnie, chociaż sprawdzamy poprawność konstrukcji.

Prosimy o nadsyłanie własnych projektów z modelami (do zwrotu). Do artykułu należy dołączyć podpisane **oświadczenie, że artykuł jest własnym opracowaniem autora i nie był dotychczas nigdzie publikowany.** Honorarium za publikację w tym dziale wynosi 250,- zł (brutto) za 1 stronę w EP. Przesyłanych tekstów nie zwracamy. Redakcja zastrzega sobie prawo do dokonywania skrótów.

„Inteligentne” tylne światło do roweru z sygnalizacją stopu

Mikrokontrolery stały się tak tanie, że można je bez uszczuplania kieszeni stosować prawie do wszystkiego. Choćby jako „inteligentne” sterowniki oświetlenia do roweru.



Projekt 108

Układ, który prezentuję, ma wiele zalet w porównaniu z typowymi (czytaj: sklepowymi) gadżetami tego rodzaju. Wśród najważniejszych można wymienić:


1. 6 efektów błyskania diod LED wraz z funkcją sygnalizowania zatrzymania, co zwiększa bezpieczeństwo rowerzysty.
2. Włączanie, wyłączenie i zmiana efektu następuje za

pomocą jednego przycisku. Po wyłączeniu, chociaż w świetle jest bateria, nie pobiera ona prądu.

Moim celem było udoskonalenie posiadanej tylnej lampki rowerowej. Przed przebudową miała 5 diod świecących, które migały podczas jazdy, a po wciśnięciu hamulca zapalały się wszystkie naraz. Wykorzystałem z niej tylko obudowę i czujnik na linkę

hamulcową. Pomimo że układ jest prosty (schemat elektryczny pokazano na rys. 1), godnym uwagi jest sposób sterowania funkcjami i włączaniem za pomocą jednego przycisku.

W układzie zastosowano mikrokontroler AT89C2051, w którego pamięci znajduje się program pokazany na list. 1. Wydajność prądowa portów mikrokontrolera jest na



Rys. 1

List. 1. Program sterujący pracą mikrokontrolera w sterowniku lampy

```

; Program został napisany przez Piotra Andryszczaka.
; Pracuje z kwarcem ok. 12MHz. Poszczególne efekty
; zostały napisane tak (za pomocą "clr" i "setb"),
; aby można było łatwiej je modyfikować.
CPU atmel.def

    org 00h
    ajmp start

    org 03h
    ajmp przycisk

    org 13h
    ajmp stop

START:
    mov P1,#FFh
    mov P3,#FFh
    mov R2,#00
    nop
    nop
    mov A,P3
    jb Acc.3,dalej
    mov P1,#FFh
    clr P1.1      ; Jesli STOP spowodowal wlaczenie
    clr P1.0      ; zasilania to diody wyswietlaja:
    clr P1.4      ;.***. *****
    mov B,#5      ; Czas opoznienia
    acall op
    clr P1.2
    clr P1.3

aa:
    ajmp aa
przycisk:
    inc R2
    reti
dalej:
    mov IE,#10000101b
    mov IP,#00000101b
    setb IT0
    setb IT1
    clr P3.0
efekt1:
    mov P1,#FFh
    cjne R2,#1,efekt2
    clr P1.2
    acall opz
    clr P1.1
    acall opz
    clr P1.0
    acall opz
    clr P1.4
    acall opz
    clr P1.3
    acall opz
    cjne R2,#1,efekt2
    setb P1.2
    acall opz
    setb P1.1
    acall opz
    setb P1.0
    acall opz
    setb P1.4
    acall opz
    setb P1.3
    acall opz
    ajmp efekt1
efekt2:
    mov P1,#FFh
    cjne R2,#2,efekt3
    clr P1.2
    clr P1.3
    mov B,#2
    acall op
    setb P1.2
    setb P1.3
    clr P1.1
    clr P1.4
    acall opz
    cjne R2,#2,efekt3
    acall opz
    setb P1.1
    setb P1.4
    clr P1.0
    mov B,#2
    acall op
    setb P1.0
    ajmp efekt2
efekt3:
    cjne R2,#3,efekt4
    mov P1,#00h
    acall opz
    mov P1,#FFh
    acall opz
    cjne R2,#3,efekt4
    mov B,#2
    acall op
    ajmp efekt3
efekt4:
    mov P1,#FFh

    cjne R2,#4,efekt5
    clr P1.2      ;LED 1
    mov B,#2
    acall op
    setb P1.2
    clr P1.3      ;LED 5
    mov B,#2
    acall op
    setb P1.3
    cjne R2,#4,efekt5
    clr P1.1      ;LED 2
    mov B,#2
    acall op
    setb P1.1
    clr P1.4      ;LED 4
    mov B,#2
    acall op
    setb P1.4
    clr P1.0      ;LED 3
    mov B,#2
    acall op
    setb P1.0
    ajmp efekt4
efekt5:
    mov P1,#FFh
    cjne R2,#5,efekt6
    clr P1.2
    clr P1.1
    mov B,#3
    acall op
    setb P1.2
    setb P1.1
    cjne R2,#5,efekt6
    clr P1.4
    clr P1.3
    mov B,#3
    acall op
    setb P1.4
    setb P1.3
    ajmp efekt5
efekt6:
    mov P1,#FFh
    cjne R2,#6,wylacz
    clr P1.2      ;LED 1
    acall opz
    setb P1.2
    clr P1.1      ;LED 2
    acall opz
    setb P1.1
    clr P1.0      ;LED 3
    acall opz
    setb P1.0
    clr P1.4      ;LED 4
    acall opz
    setb P1.4
    cjne R2,#6,wylacz
    clr P1.3      ;LED 5
    acall opz
    setb P1.3
    clr P1.4      ;LED 4
    acall opz
    setb P1.4
    clr P1.0      ;LED 3
    acall opz
    setb P1.0
    clr P1.1      ;LED 2
    acall opz
    setb P1.1
    ajmp efekt6
wylacz:
    setb P3.0      ; wylaczenie ukkladu
stop:
    push B
    mov P1,#FFh
    acall opz
    clr P1.1
    clr P1.0
    clr P1.4
    mov B,#5
    acall op
    clr P1.2
    clr P1.3
stopx:
    mov A,P3
    jnb Acc.3,stopx
    acall opz
    mov P1,#FFh
    acall opz
    pop B
    reti
opz:
    mov B,#1
op:
    mov R0,#FFh
opl:
    mov R1,#FFh
oplx:
    nop
    djnz R1,oplx
    djnz R0,opl
    djnz B,op
    ret
END

```

tyle duża, że możemy bezpośrednio sterować diodami świecącymi. Mały pobór prądu przy dużej intensywności świecenia uzyskałem dzięki zastosowaniu diod firmy Kingbright o oznaczeniu L-793SRC-F.

Opis układu

Po włożeniu baterii (2xR6) do pojemnika układ pozostaje wyłączony, aż do momentu, gdy naciśniemy hamulec w rowerze lub przycisk włączający. Gdy to nastąpi, tranzystory T1 i T2 zaczynają przewodzić, dołączając ujemny biegun zasilania do wyprowadzenia GND mikrokon-

trolera. Wtedy, po wyzerowaniu, mikrokontroler sprawdza, czy jest naciśnięty przycisk (co jest rozpoznawane przez sprawdzenie poziomu napięcia na wyprowadzeniu P3.2). Jeśli nie, oznacza to, że załączenie nastąpiło na skutek naciśnięcia hamulca i mają się zaświecić wszystkie diody. Następuje to dwustopniowo: najpierw świecą się diody 2, 3, 4, a po chwili dwie skrajne, czyli 1 i 5. Świecą do momentu puszczenia hamulca - tranzystory przestają przewodzić i zasilanie jest odcinane od układu. Jeżeli załączenie zasilania nastąpiło przyciskiem dołączonym do wy-

prowadzenia P3.2 portu, to na wyprowadzeniu P3.0 tego portu występuje poziom niski (wpisywane jest „0“) podtrzymujący załączone tranzystory nawet po jego puszczeniu. Jednocześnie wyświetlany będzie efekt numer 1. Jeżeli teraz wciśniemy ponownie przycisk, układ przejdzie do wyświetlania kolejnego efektu, a jeśli jest to ostatni z listy, to na wyprowadzeniu P3.0 wpisana zostanie jedynka i układ wyłączy się. Po wciśnięciu hamulca w czasie normalnej pracy diody świecą się, jak już wcześniej wspomniano.

Piotr Andryszczak

WYKAZ ELEMENTÓW

Rezystory

R1: 51kΩ
R2: 4,7kΩ
R3...R7: 75Ω

Kondensatory

C1, C2: 100nF
C3, C7: 220nF
C4, C5: 30pF
C6: 470μF/16V

Półprzewodniki

D1...D3: 1N4148
D4...D8: diody LED (najlepiej Kingbright typ L-793SRC-F)

T1: BC557

T2: BC547

U1: AT89C2051

(zaprogramowany)